

Einführung in die Computer-Algebra

Teil 1 : Erste Schritte mit *Mathematica*

Informatik I – Mathematik mit dem Computer

Prof. Dr. Alfred Schreiber

Institut für Mathematik und ihre Didaktik · Universität Flensburg

Literatur, etc.

Stephen Wolfram:

Das *Mathematica* Buch. Die offizielle Dokumentation.

Addison-Wesley 1997 (und weitere Versionen)

Marie-Luise Herrmann:

Mathematica. Eine beispielorientierte Einführung.

Addison-Wesley 1997

Roman. E. Maeder:

Computer Science with *Mathematica*.

Cambridge University Press 2000

Weitere Quellen:

- Website und Links zu dieser Vorlesung
- Computer im Mathematikunterricht (aus: Grundzüge der Mathematikdidaktik)
- Homepage von Wolfram Research, Inc. (www.wolfram.com)
- *MathReader* (erlaubt das Lesen von Notebooks)

Menüs, Paletten, Notebooks, Einstellungen

File

Save
Save As ...
Save As Special ...
Palettes › Auswahl diverser Paletten

Notebooks

Hierarchischer Aufbau (Zellen)
Typen von Zellen: Text, Input, Output, etc.

Einstellungen (Preferences ... im Menü Edit)

Notebook Options > Display Options > Window Toolbars
Cell Options > Display Options > ShowGroupOpenCloseIcon (True!)
Cell Options > New Cell Defaults > DefaultNewCellStyle (z.B. "Text" oder "Input")

Elementares numerisches Rechnen

■ Termeingabe in einer Input-Zelle

Bestätigen durch SHIFT + ENTER oder ENTER (Ziffernblock)

7 + 15

22

■ Rückgriff auf das letzte Ergebnis %

% * 3 / 11

6

■ Runde Klammern dienen der Gliederung (Rangfolge)

```
1234 * (35 - 3^10)
```

```
- 72823276
```

■ Große Zahlen

```
2^1000
```

```
107150860718626732094842504906000181056140481170553360744375038837035105  
51124936122493198378815695858127594672917553146825187145285692314043596  
84577574698574803934567774824230985421074605062371141877954182153046478  
4983581941267398767559165543946077062914571196477686542167660429831652  
624386837205668069376
```

■ Numerische Näherungswerte

```
N[%]
```

```
1.07151 × 10301
```

■ Argumentklammern sind eckig [...]

```
Mod[2^32 + 1, 641]
```

```
0
```

■ Zwei wichtige mathematische Konstanten

Die Kreiszahl π

```
Pi  
N[Pi, 30]
```

```
π
```

```
3.14159265358979323846264338328
```

Die Eulersche Zahl e

```
E  
N[E, 30]
```

```
e
```

```
2.71828182845904523536028747135
```

■ Quadratwurzel

Anfangsbuchstaben der Namen eingebauter Funktionen sind **groß** zu schreiben!

```
Sqrt[174]
```

```
 $\sqrt{174}$ 
```

Bestimmte Funktionen lassen sich in nachgestellter Schreibweise (hinter // ausführen)

```
% // N
```

```
13.1909
```

■ Fakultät-Funktion

Beispiel für die Übernahme einer verbreiteten historischen Schreibweise

```
25!
```

```
15511210043330985984000000
```

■ Binomialzahlen

Beispiel einer eingebauten Funktion von zwei Argumenten

```
Binomial[7, 3]
```

```
35
```

■ Gebrauch von Symbolen (Variablen)

■ "Buchstabenrechnung"

Symbole können undefiniert in Ausdrücken erscheinen

```
Sqrt[a]  
(a + b) ^ 2
```

```
 $\sqrt{a}$ 
```

```
 $(a + b)^2$ 
```

Auskunft über den sog. Kontext einholen

```
? a b
```

```
Global`a
```

```
Global`b
```

■ Wertzuweisung

```
a = 13
```

```
13
```

Nun ist das Symbol a definiert

```
? a
```

```
Global`a
```

```
a = 13
```

Freie Namenswahl

(**keine** Unterstriche verwenden! Am besten nur Buchstaben A, a, B, b, \dots und Ziffern $0, 1, 2, \dots$)

```
meinevar = 75
```

```
75
```

■ Gebrauch des Zeichens "="

"=" kann ähnlich wie in der Mathematik verwendet werden ...

```
b = c = a
```

```
13
```

Aber: **WARNUNG** – Eine Wertzuweisung ist **keine Gleichung**.

Die folgende Zeichenkette ist Unsinn!

```
8 = c
```

```
Set::setraw : Cannot assign to raw object 8.
```

```
13
```

■ Namen sollen "sprechen"

Ein Semikolon trennt zwei Befehle voneinander

```
radius = 5;  
kreisfläche = Pi * radius^2
```

```
25 π
```

```
% // N
```

```
78.5398
```

■ Inhalt von Symbolen löschen

```
? b
```

```
Global`b
```

```
b = 13
```

Clear löscht den Inhalt von einer oder mehreren Variablen

```
Clear[b, c]
```

Die Symbole sind aber noch bekannt (d.h. hier: im Kontext Global)

```
? b
```

```
? c
```

```
Global`b
```

```
Global`c
```

Statt Clear[a] ist auch folgende Wertzuweisung möglich

```
a = .
```

■ Bezug zu Symbolen entfernen

```
Remove[b]
```

Zu b gibt es keinen Bezug mehr

```
? b
```

```
? c
```

```
Information::notfound : Symbol b not found.
```

```
Global`c
```

■ Ersetzungsregeln

Auf Ausdrücke Transformationsregeln anwenden:

```
Ausdruck /. ls -> rs
```

Ein gewöhnlicher Ausdruck:

```
x + 2 y + z
```

```
x + 2 y + z
```

x durch a ersetzen:

```
x + 2 y + z /. x -> a
```

```
a + 2 y + z
```

a durch a-z ersetzen:

```
% /. a -> a - z
```

```
a + 2 y
```

Relationale und logische Operatoren

■ Kleiner(gleich), Größer(gleich)

```
32 < 1 + 6
```

```
False
```

```
45 > 3
```

```
True
```

```
40 ≤ 41
```

```
True
```

■ Gleich, Ungleich

```
2 + 3 == 5
```

```
True
```

Negation von Gleich: `!=`

```
2 + 3 ≠ 5
```

```
False
```

■ Und, Oder

Und-Verknüpfung

```
(Pi > E) && (Sqrt[6] > 2.5)
```

```
False
```

Oder-Verknüpfung (einschließend)

```
(Pi > E) || (Sqrt[6] > 2.5)
```

```
True
```

■ Wenn ... dann ... sonst

```
If[Pi > E, 1, 0]
```

```
1
```

```
If[2 + 2 ≠ 4, "Huh!", "Gottseidank!"]
```

```
Gottseidank!
```

Listen

■ Was ist eine Liste? - Schreibweise

Listen (geordnete Sammlung beliebiger Elemente)

```
{5, x, -Pi, 5, "Hallo"}
```

```
{5, x, -π, 5, Hallo}
```

Eine Liste ist keine Menge

```
{5, x, -Pi, 5, "Hallo"} == {5, x, -Pi, "Hallo"}
```

```
False
```

Auf die Reihenfolge komm es an

```
{5, x, -Pi, 5, "Hallo"} == {5, 5, x, "Hallo", -Pi}
```

```
False
```

Warum kann *Mathematica* den nachstehenden Test auf Gleichheit nicht entscheiden?

```
{5, x, -Pi, 5, "Hallo"} == {5, 5, -Pi, x, "Hallo"}
```

```
{5, x, -π, 5, Hallo} == {5, 5, -π, x, Hallo}
```

■ Erzeugung

Eine Liste der ersten 10 Quadratzahlen

```
Table[k^2, {k, 1, 10}]
```

```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

■ Wertzuweisungen

Es ist vorteilhaft, Listen einen Namen zu geben

```
qzliste = Table[k^2, {k, 1, 10}];
```

Elementweise Mehrfach-Wertzuweisung

```
{a, b, c} = {7, -1, 23};
```

```
b
```

```
-1
```

■ Listen als Elemente von Listen

Listen können selbst wieder Element einer Liste sein usf.

Beispiel: Liste aller Wertepaare (x, x^2)

```
qzWerteTab = Table[{x, x^2}, {x, 5, 9}]
```

```
{{5, 25}, {6, 36}, {7, 49}, {8, 64}, {9, 81}}
```

■ Darstellungsformen

Liste in Form einer Matrix darstellen

```
MatrixForm[qzWerteTab]
```

```

$$\begin{pmatrix} 5 & 25 \\ 6 & 36 \\ 7 & 49 \\ 8 & 64 \\ 9 & 81 \end{pmatrix}$$

```

Liste in Form einer Tabelle darstellen (MatrixForm und TableForm lassen sich mittels // anhängen)

```
qzWerteTab // TableForm
```

```
5    25
6    36
7    49
8    64
9    81
```

Sog. Optionen sorgen für eine Beschriftung

```
TableForm[qzWerteTab, TableHeadings → {None, {"x", "x2"}}]
```

```
x    x2
5    25
6    36
7    49
8    64
9    81
```

■ Länge einer Liste

```
Length[qzWerteTab]
```

```
5
```

■ Zugriff auf Elemente

Aus der zuvor erzeugten Liste ...

```
qzliste
```

```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

wird das Element Nr. 7 herausgeholt:

```
Part[qzliste, 7]
```

```
49
```

Alternative Schreibweise (mit eckigen Doppelklammern)

```
qzliste[[7]]
```

```
49
```

Die Funktion **Part**[] bietet mehr Möglichkeiten:

```
Part[qzliste, {7, 9}]
```

```
{49, 81}
```

Symbolisches Rechnen

■ Ausmultiplizieren

```
b = (1 + x) ^ 20
```

```
(1 + x) ^ 20
```

```
Expand[b]
```

```
1 + 20 x + 190 x^2 + 1140 x^3 + 4845 x^4 + 15504 x^5 + 38760 x^6 + 77520 x^7 +
125970 x^8 + 167960 x^9 + 184756 x^10 + 167960 x^11 + 125970 x^12 + 77520 x^13 +
38760 x^14 + 15504 x^15 + 4845 x^16 + 1140 x^17 + 190 x^18 + 20 x^19 + x^20
```

■ Reduzieren

```
Factor[-27 - 54 u - 27 u^2 + 54 x + 162 u x + 108 u^2 x + 45 x^2 - 72 u x^2 - 144 u^2 x^2 - 116 x^3 -
160 u x^3 + 64 u^2 x^3 - 16 x^4 + 128 u x^4 + 64 x^5 + 162 y^4 + 324 u y^4 + 162 u^2 y^4 - 108 x y^4 -
540 u x y^4 - 432 u^2 x y^4 - 414 x^2 y^4 - 288 u x^2 y^4 + 288 u^2 x^2 y^4 + 144 x^3 y^4 + 576 u x^3 y^4 +
288 x^4 y^4 - 324 y^8 - 648 u y^8 - 324 u^2 y^8 - 216 x y^8 + 216 u x y^8 + 432 u^2 x y^8 + 540 x^2 y^8 +
864 u x^2 y^8 + 432 x^3 y^8 + 216 y^12 + 432 u y^12 + 216 u^2 y^12 + 432 x y^12 + 432 u x y^12 + 216 x^2 y^12]
```

```
(1 + u + x)^2 (-3 + 4 x + 6 y^4)^3
```

Alle Variableninhalte löschen (um im Folgenden keine Überraschungen zu erleben)

```
Clear["Global`*"]
```

■ Vereinfachen (allgemein)

```
Simplify[(a - b)^2 - (a + b)^2]
```

```
-4 a b
```

■ Summen berechnen

```
Sum[k^2, {k, 1, n}]
```

```
 $\frac{1}{6} n (1 + n) (1 + 2 n)$ 
```

■ Gleichungen lösen

Lösung wird als Menge von Ersetzungsregeln ausgegeben

```
lsg = Solve[x^2 + a x == 7, x]
```

```
 $\left\{ \left\{ x \rightarrow \frac{1}{2} (-a - \sqrt{28 + a^2}) \right\}, \left\{ x \rightarrow \frac{1}{2} (-a + \sqrt{28 + a^2}) \right\} \right\}$ 
```

Übergang zur Lösungsmenge ...

```
x /. lsg
```

```
 $\left\{ \frac{1}{2} (-a - \sqrt{28 + a^2}), \frac{1}{2} (-a + \sqrt{28 + a^2}) \right\}$ 
```

... oder direkt auf eine einzelne Lösung zugreifen:

```
x /. lsg[[1]]
```

```
 $\frac{1}{2} (-a - \sqrt{28 + a^2})$ 
```

Beispiel eines linearen Gleichungssystems

```
Solve[{a x + b y == c, d x + e y == f}, {x, y}]
```

$$\left\{ \left\{ x \rightarrow -\frac{-c e + b f}{-b d + a e}, y \rightarrow -\frac{-c d + a f}{b d - a e} \right\} \right\}$$

An der Grenze algebraischer Lösbarkeit:

```
x /. Solve[x^4 + 3 x - 1 == 0, x][[1]]
```

$$\frac{1}{2} \sqrt{-4 \left(\frac{2}{3 (81 + \sqrt{7329})} \right)^{1/3} + \frac{\left(\frac{1}{2} (81 + \sqrt{7329}) \right)^{1/3}}{3^{2/3}}} -$$

$$\frac{1}{2} \sqrt{\left(4 \left(\frac{2}{3 (81 + \sqrt{7329})} \right)^{1/3} - \frac{\left(\frac{1}{2} (81 + \sqrt{7329}) \right)^{1/3}}{3^{2/3}} - \right.$$

$$\left. \frac{6}{\sqrt{-4 \left(\frac{2}{3 (81 + \sqrt{7329})} \right)^{1/3} + \frac{\left(\frac{1}{2} (81 + \sqrt{7329}) \right)^{1/3}}{3^{2/3}}}} \right)}$$

Funktionen

■ Eigene Funktionen definieren

Argument x als Muster x_ (nur auf der linken Seite)

Definitionsoperator :=

```
f[x_] := (x / 2) (x^3 - 5 x^2 + 1)
```

Weitere Handhabung wie in der Mathematik üblich:

```
f[a + 3] // Expand
```

$$-\frac{51}{2} - 13 a + \frac{9 a^2}{2} + \frac{7 a^3}{2} + \frac{a^4}{2}$$

■ Definitionen anzeigen lassen

```
? f
```

```
Global`f
```

```
f[x_] :=  $\frac{1}{2} x (x^3 - 5 x^2 + 1)$ 
```

■ Funktionen mit mehreren Argumenten

Früher definierte Funktionen dürfen (natürlich) benutzt werden

```
g[u_, v_] := u * f[v] + v^2 / u
```

```
g[a, b]
```

```
 $\frac{b^2}{a} + \frac{1}{2} a b (1 - 5 b^2 + b^3)$ 
```

■ Nicht-Benötigtes löschen

Unbedingt zu empfehlen!

Eventuell nur die Definition löschen ...

```
Clear[g]
```

... oder nicht mehr benötigte Funktionssymbole vollständig entfernen:

```
Remove[f, g]
```

```
f[4]
```


```
f[4]
```

Zeichnen ebener Schaubilder

■ Einfacher Funktionsgraph


Der Plot-Befehl in der Grundform:

```
Plot[3 x - x^2, {x, -5, 5}];
```




Mit Optionen lässt sich ein Schaubild verfeinern, z.B. Achsenbeschriftung und Strichdicke:

```
Plot[3 x - x^2, {x, -5, 5},  
  AxesLabel -> {"x-Achse", "y-Achse"}, PlotStyle -> {Thickness[0.01]}];
```



■ Mehrere Kurven in einem Schaubild


```
Plot[{x, Sqrt[x], Log[x]}, {x, 0, 10}, AspectRatio -> 0.5, PlotRange -> {-3, 3}];
```



■ Funktionsscharen

Die Schar der Exponentialfunktionen

```
schar = Table[a^x, {a, 1, 3, 0.25}];  
Plot[Evaluate[schar], {x, -5, 5}, PlotRange -> {0, 10}];
```



■ Optionen des Plot-Befehls

So informiert man sich über die (voreingestellten) Optionen eines Befehls:

Options[Plot]

```
{AspectRatio →  $\frac{1}{\text{GoldenRatio}}$ , Axes → Automatic, AxesLabel → None,
  AxesOrigin → Automatic, AxesStyle → Automatic, Background → Automatic,
  ColorOutput → Automatic, Compiled → True, DefaultColor → Automatic,
  Epilog → {}, Frame → False, FrameLabel → None, FrameStyle → Automatic,
  FrameTicks → Automatic, GridLines → None, ImageSize → Automatic,
  MaxBend → 10., PlotDivision → 30., PlotLabel → None, PlotPoints → 25,
  PlotRange → Automatic, PlotRegion → Automatic, PlotStyle → Automatic,
  Prolog → {}, RotateLabel → True, Ticks → Automatic,
  DefaultFont := $DefaultFont, DisplayFunction := $DisplayFunction,
  FormatType := $FormatType, TextStyle := $TextStyle}
```

Die Bedeutung der Optionen erhält man über die Hilfe.


(Im Output die Schreibmarke auf das Optionswort setzen und anschließend F1-Taste drücken).

■ Datenlisten zeichnen

Beispiel-Daten:

```
daten = Table[Mod[k (k - 1), 3 k + 1], {k, 1, 20}];
```

```
ListPlot[daten, PlotStyle → {PointSize[0.015]}];
```




Mit der Option `PlotJoined -> True` können die Punkte verbunden werden.

■ Kurven in Parameterdarstellung

Eine Ellipse (deren Schaubild ist nicht der Graph einer Funktion):

```
ParametricPlot[{2 Cos[t], Sin[t]}, {t, 0, 2*Pi}];
```




3D-Grafik

■ Dreidimensionales Linien-Zeichnen

Eine Schraubenlinie in Parameterdarstellung

```
ParametricPlot3D[{Cos[t], Sin[t], t/3}, {t, 0, 15}];
```




■ Flächen im Raum

Eine Fläche benötigt im Raum 2 Parameter.

Der folgende Torus wird dadurch erzeugt, dass mit der Variation von u ein Kreis entsteht und dieser zusätzlich mit der Variation von t auf einer Kreisbahn um die z-Achse rotiert:

```
ParametricPlot3D[  
  {Cos[t] * (3 + Cos[u]), Sin[t] * (3 + Cos[u]), Sin[u]}, {t, 0, 2 Pi}, {u, 0, 2 Pi}];
```




■ Polyeder

Zunächst ist ein sog. *Mathematica*-Paket zu laden:

```
<< Graphics`Polyhedra`
```


Darstellung eines Dodekaeders:

```
Show[Graphics3D[Dodecahedron[], ViewPoint -> {1, 1, 0}]];
```



Wir entfernen die obenliegende Seitenfläche:

```
dd = Drop[Dodecahedron[], {1}];  
Show[Graphics3D[dd]];
```



Animation

■ *Mathematica*-Package für Animation

Folgender Befehl lädt das Animationspaket:

```
<< Graphics`Animation`
```

■ Ebene Grundform


`Animate[]` übernimmt einen Plot-Befehl, der eine Schar von Objekten erzeugt. Die Anweisungsliste für den Scharparameter wird hinzugefügt:

```
Animate[Plot[Sin[n x], {x, 0, 2 Pi}, Axes -> False], {n, 1, 6, 0.5}];
```

■ Animation einer räumlichen Figur

Zunächst wird ein Objekt `g` erzeugt:

```
g = ParametricPlot3D[{x, Cos[t] Sin[x], Sin[t] Sin[x]},  
  {x, -Pi, Pi}, {t, 0, 2Pi}, Axes -> False, Boxed -> False];
```



Nun wird `g` mittels `SpinShow` animiert (10 Phasen). Aufgrund der Symmetrie ist nur eine Halbdrehung erforderlich:

```
SpinShow[g, Frames -> 10, SpinRange -> {0 Degree, 180 Degree}];
```